AAE 706

Optimization and Equilibrium

Thomas F. Rutherford

Department of Agricultural and Applied Economics
University of Wisconsin, Madison

February 15, 2023

Calibration Exercise #1

Suppose that irregardless of relative prices, Suzy always has one soda before and one soda after eating an ice cream. What utility function is consistent with these choices? Write down demand functions which could extrapolate her optimal choices to any expenditure (m) and prices (p_1 and p_2).

Calibration Exercise #1: Answer

Perfect complement preferences have the form:

$$U(x_1, x_2) = \min(\frac{x_1}{a_1}, \frac{x_2}{a_2})$$

in which the ratio $\frac{a_1}{a_2}$ determines the ratio in which goods 1 and 2 are consumed. In the present example, we have:

$$U(x_1, x_2) = \min(x_1, \frac{x_2}{2})$$

and demand functions given by:

$$x_1 = \frac{Y}{p_1 + 2p_2}$$

and

$$x_2 = 2\frac{Y}{p_1 + 2p_2}$$

Calibration Exercise #2

When Joe gets to the bar, he always asks about the price of peanuts and the price of beer. If two beers cost less than one bag of peanuts, he spends all of his money on beer. Otherwise he buys peanuts. What utility function is consistent with these choices? Write down demand functions which could extrapolate her optimal choices to any expenditure (m) and prices (p_1 and p_2).

Calibration Exercise #2: Answer

General *perfect substitues preferences* have the form:

$$U(x_1, x_2) = a_1 x_1 + a_2 x_2$$

in which the ratio $\frac{a_1}{a_2}$ represents the marginal rate of substitution of good 1 for good 2. The demand functions for these preferences are given by:

$$x_1 = \begin{cases} 0 & \text{when } \frac{p_1}{p_2} > \frac{a_1}{a_2} \\ \frac{M}{p_1} & \text{otherwise} \end{cases}$$

$$x_2 = \begin{cases} 0 & \text{when } \frac{p_1}{p_2} < \frac{a_1}{a_2} \\ \frac{M}{p_2} & \text{otherwise} \end{cases}$$

Paul Samuelson in 1950



Paul Samuelson

- University of Chicago (B.A.) – Harvard University (Ph.D.),
- Enrolled college at age 16
- Full professor at age 32
- First American to win the Nobel Memorial Prize in Economic Sciences: "[Samuelson] has done more than any other contemporary economist to raise the level of scientific analysis in economic theory."
- Recruited numerous Nobel laureates at MIT: Robert M. Solow, Paul Krugman, Franco Modigliani, Robert C. Merton and Joseph E. Stiglitz.

A Theory which is both True and Nontrivial

Stanislaw Ulam once challenged Samuelson to name one theory in all of the social sciences which is both true and nontrivial.

A Theory which is both True and Nontrivial

Stanislaw Ulam once challenged Samuelson to name one theory in all of the social sciences which is both true and nontrivial.
Several years later, Samuelson responded with David Ricardo's theory of comparative advantage:

> *That it is logically true need not be argued before a mathematician; that is not trivial is attested by the thousands of important and intelligent men who have never been able to grasp the doctrine for themselves or to believe it after it was explained to them.*

Activity Analysis

When there are a discrete set of production technologies, each characterized by a marginal cost and a capacity, the supply curve becomes a step function corresponding to the sorted sequence of plant capacities.

Consider a market in which the commodity is supply by the following four technologies:

	c_j	k_j
a	2	2
b	5	2
c	7	4
d	10	∞

Activity Analysis Supply Curve



Consider a market equilibrium when there are multiple discrete supply technologies. As in the conventional continuous Marshallian model, the equilibrium price and quantity is defined by the intersection of the supply and demand schedules:



Market Equilibrium and Social Surplus

A convenient property of the competitive market allocation is that it *maximizes* social surplus, as illustrated in this figure:



Constrained Optimization Approach

Let $Q_t \geq 0$ denote output from technology t, P denote the equilibrium price, PS and CS denote producer and consumer surplus. The market equilibrium then solves:

$$\max PS + CS$$

subject to:

- Market supply equals technology output:

$$S = \sum_t Q_t$$

- Market equilibrum price is on the demand curve:

$$P = 10 - \frac{5}{6} S$$

- Producer surplus is the area below the market price and above the cost of production: $PS = \sum_t (P - c_t) Q_t$
- Consumer surplus is the area under the demand curve: $CS = \frac{(10-P)}{2} S$

Geometric Interpretation of the Equilibrium



GAMS Code – Sets and Data

```
$title  surplus maximization and market equilibrium

set     t /a,b,c,d/;

table   tech    Technology
        cost    cap
a       2       2
b       5       2
c       7       4
d       10      inf;

parameter       c(t)    Cost by technology;
c(t) = tech(t,"cost");
```

GAMS Code – Variable Declaration

```
nonnegative variables   P,PS,CS,s,Q(t);
free variable           obj;
equations               price, supply, psurplus, csurplus, objective

price..          P =e= 10 - S*10/6;

supply..         S =e= sum(t, Q(t));

psurplus..       PS =e= sum(t, (P-c(t))* Q(t));

csurplus..       CS =e= (10 - P)*S/2;

objective..      OBJ =e= CS + PS;

Q.UP(t) = tech(t,"cap");

model equil /all/;
solve equil using nlp maximizing OBJ;
```

GAMS Listing File

```
                 LOWER        LEVEL        UPPER      MARGINAL
---- VAR P        .           5.0000     +INF           .
---- VAR PS       .           6.0000     +INF           .
---- VAR CS       .           7.5000     +INF           .

---- VAR Q
a                 .           2.0000       2.0000      3.0000
b                 .           1.0000       2.0000      EPS
c                 .            .           4.0000     -2.0000
d                 .            .         +INF         -5.0000

---- VAR obj     -INF          13.5000   +INF           .
```

Return to the Electricity Investment Model

- Sets

  s Load segments

  j Generating units, e.g. existing capacity, new investment options

  i Demand categories, e.g. residential, commercial, industrial

  f Fuel types, e.g. hard coal, soft coal, natural gas, uranium

Market Data

h_s Segment durations, hours

$\bar{p}_s$, $\bar{D}_{is}$, ϵ_{is} Demand characteristics as might be represented by representative price-quantity paris and elasticities of demand (price expressed in € per KW, demand in KW and elasticity is dimensionless)

Unit Level Data

- ϕ_{fj} Heat rates describing input fuel requirements per unit generation (PJ per KWH)
- $\bar{K}_j$ Capacities of existing generating units, TW
- c_f Fuel costs (€ per PJ)
- α_{js} Average availability factor for generating units, reflecting need for repair and intermittency of renewable energy sources (dimensionless)
- r_j^K Rental price of *new* generating capacity, (€ per KW per year), typically computed on the basis of capital cost, depreciation rate, capital cost and fixed maintenance and operating costs:

$$r_j^K = \begin{cases} p_j^K(r+\delta) + c_j^M & \text{New plants} \\ c_j^M & \text{Extant plants} \end{cases}$$

- p_j^M Variable maintenance and operating costs, (€ per KWH)

Variables which Determine an Equilibrium

- Primal Variables : quantities

  X_{js} Generation and dispatch

  K_j Generating utilization (extant and new vintage)

- Dual Variables : prices

  p_s Wholesale prices by load segment

  π_{js} Profit margins

  μ_j Shadow price on installed (extant) capacity

Primal Equilibrium Conditions

- Aggregate demand:

$$D_s = \sum_i \bar{D}_{is} \left(1 - |\epsilon_{is}|(p_s/\bar{p}_s - 1)\right) \quad \perp \quad D_s$$

Primal Equilibrium Conditions

- Aggregate demand:

$$D_s = \sum_i \bar{D}_{is} \left(1 - |\epsilon_{is}|(p_s/\bar{p}_s - 1)\right) \quad \perp \quad D_s$$

- Market clearance:

$$D_s = \sum_j X_{js} \quad \perp \quad p_s$$

Primal Equilibrium Conditions

- Aggregate demand:

$$D_s = \sum_i \bar{D}_{is} \left(1 - |\epsilon_{is}|(p_s/\bar{p}_s - 1)\right) \quad \perp \quad D_s$$

- Market clearance:

$$D_s = \sum_j X_{js} \quad \perp \quad p_s$$

- Feasibility of generation:

$$\alpha_{js} K_j \geq X_{js} \geq 0 \quad \perp \quad \pi_{js} \geq 0$$

Primal Equilibrium Conditions

- Aggregate demand:

$$D_s = \sum_i \bar{D}_{is} \left(1 - |\epsilon_{is}|(p_s/\bar{p}_s - 1)\right) \quad \perp \quad D_s$$

- Market clearance:

$$D_s = \sum_j X_{js} \quad \perp \quad p_s$$

- Feasibility of generation:

$$\alpha_{js} K_j \geq X_{js} \geq 0 \quad \perp \quad \pi_{js} \geq 0$$

- Capacity:

$$\bar{K}_j \geq K_j \quad \perp \quad \mu_j \geq 0$$

Dual Equilibrium Conditions

- Profitability – arbitrage in dispatch:

$$\pi_{js} + p_j^M + \sum_f c_f \phi_{fj} \geq p_s \quad \perp \quad X_{js} \geq 0$$

Dual Equilibrium Conditions

- Profitability – arbitrage in dispatch:

$$\pi_{js} + p_j^M + \sum_f c_f \phi_{fj} \geq p_s \quad \perp \quad X_{js} \geq 0$$

- Profitability – arbitrage in investment:

$$r_j^K + \mu_j \geq \sum_s h_s \alpha_{js} \pi_{js} \quad \perp \quad K_j \geq 0$$

Integrability: Equilibrium Allocation = Optimal Allocation

$$\begin{aligned}\max \quad & \textstyle\sum_{i,s} \bar{p}_s D_{is} \left(1 + 1/|\epsilon_{is}|(1 - \frac{D_{is}}{2\bar{D}_{is}})\right) \\ & - \textstyle\sum_{sj} X_{js} h_s \left(\sum_f c_f \phi_{jf} + p_j^M\right) - \sum_j K_j r_j^K\end{aligned}$$

subject to:

$$\sum_{is} D_{is} = \sum_j X_{js}$$

$$\alpha_{js} K_j \geq X_{js} \geq 0$$

$$\bar{K}_j \geq K_j$$

$$K_j \geq 0$$

Electricity Dispatch GAMS Code: LP Model

```
variables       TOTCOST Objective function (dispatch cost);

nonnegative
variables       Y(j,s)  Dispatch;

equations       costdef, demand;

costdef..       TOTCOST =e= sum((j,s), mc(j)*Y(j,s));

demand(s)..     sum(j, Y(j,s)) =e= qref(s);

model mincost /costdef, demand/;

Y.UP(j,s) = supply(j,"cap");

solve mincost minimizing TOTCOST using LP;
```

GAMS Code for the Equilibrium Model

```
parameter       pref(s)         Reference price,
                dref(i,s)       Reference demand;

pref(s) = demand.m(s);
dref(i,s) = qref(s) * load(s,i);

parameter       epsilon(i)      Elasticity of demand /rsd 0.1,  com 0.2,  ind 0.5/;

nonnegative
variables       D(i,s)  Aggregate demand,
                PI(j,s) Shadow price on capacity,
                P(s)    Market price;

equations aggdemand, supplydemand, profit, capacity;

aggdemand(i,s)..        D(i,s) =e= dref(i,s) * (1 - epsilon(i)*(P(s)/pref(s)-1));

supplydemand(s)..       sum(j,Y(j,s)) =e= sum(i,D(i,s));

profit(j,s)..           mc(j) + PI(j,s) =g= P(s);

capacity(j,s)..         cap(j) =g= Y(j,s);

model equil /aggdemand.D, supplydemand.P, profit.Y, capacity.PI/;

Y.UP(j,s) = inf;  P.L(s) = pref(s);  D.L(i,s) = dref(i,s);  PI.L(j,s) = -Y.M(j,s);

equil.iterlim = 0;
SOLVE equil USING mcp;
```

Electricity Dispatch GAMS Code: Equilvalent QP Model

```
variable        SURPLUS         Sum of consumer and producer surplus
                K(j)            Capacity of technology j;

equation        surplusdef      Defines the surplus;

surplusdef..    SURPLUS =e= sum((i,s), pref(s)*D(i,s) *
                                (1 + 1/epsilon(i) * (1 - D(i,s)/(2*dref(i,s)))))
                         -  TOTCOST;

MODEL SAMUELSON /surplusdef, supplydemand, costdef, capacity/;

K.FX(j) = supply(j,"cap");
Y.UP(j,s) = +inf;

SOLVE samuelson USING nlp MAXIMIZING surplus;
```

Homework Assignment for Monday

The *isoelastic* demand function is an alternative functional form:

$$d(p) = ap^b$$

1. Derive values of a and b which produce a demand function which is locally consistent with the following *linear* demand curve at $p = \bar{p}$:

$$d(p) = \bar{d}\left(1 - |\epsilon|\left(\frac{p}{\bar{p}} - 1\right)\right)$$

2. Formulate a representation of the isoelastic demand based on $\bar{q}$, $\bar{p}$ and ϵ rather than a and b.
3. Produce MCP and NLP models with iso-elastic demand, and demonstrate that these are calibrated.
4. Impose a supply shock (a phase out of coal generation) and compare results from the linear and isoelastic models.

Spatial Price Equilibrium

i	Supply nodes
j	Demand nodes
c_{ij}	Unit shipment costs
μ_i	Unit (marginal) production cost
$\bar{S}_i$	Supply limit (upper bound)
$\bar{D}_j$	Demand quantity

Least Cost Production and Distribution

$$\min \sum_i \mu_i S_i + \sum_{i,j} c_{ij} X_{ij}$$

subject to:

$$\begin{aligned} S_i \geq & \quad \textstyle\sum_j X_{ij} \\ \sum_i X_{ij} \geq & \quad D_j \end{aligned}$$

$$D_j = \bar{D}_j, \quad S_i \leq \bar{S}_i$$

GAMS Code

```
$title  A Calibrated Spatial Price Equilibrium Model

$ontext

We first formulate a linear programming model which minimizes
the cost of production and distribution on a transportation
network with supply nodes and demand nodes.  Using the primal
and dual values from the LP model we calibrate an economic
equilibrium model with price elastic demand and supply for
which the reference equilibrium corresponds precisely to the
LP optimum.

$offtext

*       Generate a random instance of the problem:

set     i       Supply nodes /1*5/
        j       Demand nodes /1*5/;


parameter       d0(j)   Demands
                s0(i)   Supply
                mu(i)   Marginal cost of production,
                c(i,j)  Transport cost;

c(i,j) = uniform(0,1);
d0(j)  = round(uniform(1,100));
s0(i)  = round(uniform(1,200));
mu(i)  = uniform(0.5,1.5);
```

GAMS Code (cont)

```
*       Here I illustrate the lazy way to declare variables.  When
*       a variable is declared with no arguments, the dimensionality
*       is inferred at the first use and the domains are assumed
*       to be the universe, e.g. X(*,*).

*       The disadvantage of this approach is that domain errors
*       may be undetected and difficult to trace. It is a good idea
*       to use explicit domain whereever possible:

nonnegative variables   X,S,D;

free variable   TOTCOST         Objective function;

equations       objdef, supply, demand;

objdef..        TOTCOST =e= sum((i,j), c(i,j) * X(i,j)) + sum(i, mu(i)*S(i));

*       Orient both equations as >= so that the Lagrange multipliers
*       are non-negative:

supply(i)..     S(i) =g= sum(j, X(i,j));

demand(j)..     sum(i, X(i,j)) =g= D(j);

model transport /all/;

*       Fix demand and place an upper bound on supply in order
*       that the marginal cost of supply is included in the
*       shadow prices at the equilibrium point:

S.UP(i) = s0(i);  D.FX(j) = d0(j);

solve transport using LP MINIMIZING TOTCOST;
```

LP Solution

Formulated as a capacity-constrained supply with constant marginal cost, the shadow prices at supply and demand nodes reflect both the production and transportation costs:

```
---- EQU supply

        LOWER          LEVEL          UPPER         MARGINAL

1         .              .            +INF            1.3821
2         .              .            +INF            1.3298
3         .              .            +INF            1.2227
4         .              .            +INF            1.1282
5         .              .            +INF            1.2468

---- EQU demand

        LOWER          LEVEL          UPPER         MARGINAL

1         .              .            +INF            1.5539
2         .              .            +INF            1.2878
3         .              .            +INF            1.3783
4         .              .            +INF            1.3969
5         .              .            +INF            1.3534
```

Calibrated Supply and Demand Functions



Calibrated Supply and Demand Functions

Given the following *data*:

- $\bar{q}$ Reference quantity supplied (and demanded)
- $\bar{p}$ Reference demand price
- $\bar{\mu}$ Reference supply price
- ϵ Magnitude of the price elasticity of demand
- η Magnitude of the price elasticity of supply

We can write the demand and supply functions as:

$$d(p) = \bar{d}\left(1 - \epsilon\left(\frac{p}{\bar{p}} - 1\right)\right)$$

and

$$s(\mu) = \bar{s}\left(1 + \eta\left(\frac{\mu}{\bar{\mu}} - 1\right)\right)$$

Mathematical Facts

1. $$\frac{\mathrm{d}}{\mathrm{d}Q}\int_{q=0}^{Q} p(q)dq = p(Q)$$

2. The first order conditions for

$$\max \sum_i f_i(S_i) + \sum_j g_j(D_j)$$

s.t.

$$\begin{aligned} S_i \geq & \sum_j X_{ij} & \perp \mu_i \\ \sum_i X_{ij} \geq & D_j & \perp p_j \end{aligned}$$

are

$$\frac{\mathrm{d}f_i(S_i)}{\mathrm{d}S_i} = -\mu_i$$

and

$$\frac{\mathrm{d}g_j(D_i)}{\mathrm{d}D_i} = p_j$$

Integrable Demand

The calibrated inverse demand function corresponding to $D_j(p_j)$ is

$$p_j(q) = \bar{p}_j \left(1 + \left(1 - q/\bar{D}_j\right)/\epsilon_j\right)$$

and the calibrated inverse supply function corresponding to $S_i(\mu_i)$ is

$$\mu_i(q) = \bar{\mu}_i \left(1 + \left(q/\bar{S}_i - 1\right)/\eta_i\right)$$

Integrating, we have consumer surplus

$$CS_j(D_j) = \int^{D_j} p_j(q)\mathrm{d}q = \bar{p}_j D_j \left(1 + \left(1 - \frac{D_j}{2\bar{D}_j}\right)/\epsilon_j\right)$$

and total cost

$$TC_i(S_i) = \int^{S_i} \mu_i(q)\mathrm{d}q = \bar{\mu}_i S_i \left(1 + \left(\frac{S_i}{2\bar{S}_i} - 1\right)/\eta_i\right)$$

The Integrated Equilibrium Model

$$\max \sum_j \underbrace{\int^{D_j} p_j(q)\mathrm{d}q}_{CS_j(D_j)} - \sum_i \underbrace{\int^{S_i} \mu_i(q)\mathrm{d}q}_{TC_i(S_i)} - \sum_{ij} c_{ij} X_{ij}$$

s.t.

$$\begin{aligned} S_i &\geq \textstyle\sum_j X_{ij} && \perp \mu_i \\ \textstyle\sum_i X_{ij} &\geq D_j && \perp p_j \\ X_{ij} &\geq 0 \end{aligned}$$

Price-Responsive Demand (QCP Formulation)

```
*       Extract the solution with fixed demand as a reference equilibrium:

parameter       muref(i)        Reference marginal cost
                pref(j)         Reference demand price
                sref(i)         Reference supply
                dref(j)         Reference demand
                epsilon(j)      Demand elasticity at the reference point;

muref(i) = supply.m(i); pref(j) = demand.m(j); sref(i) = S.L(i); dref(j) = D.L(j);

epsilon(j) = uniform(0.5, 2);


free variable   SURPLUS         Social surplus;

equation        csurplus        Social surplus with horizontal supply curves (Cs);

csurplus..      SURPLUS =e= -sum((i,j), c(i,j) * X(i,j)) - sum(i, mu(i)*S(i))

        + sum(j,  pref(j) * D(j) * (1 + (1-0.5*D(j)/dref(j)) / epsilon(j)));

model elasticdemand /supply, demand, csurplus/;

*       Remove upper and lower bounds on demand:

D.LO(j) = 0;  D.UP(j) = +inf;

solve elasticdemand using QCP maximizing SURPLUS;
```

QCP Solution

Formulated as a maximization problem, Lagrange multipliers on the supply and demand markets change sign, but they have identical magnitude as compared with the LP solution. This implies that we have "replicated the benchmark equilibrium", having removed upper and lower bounds on demand but introduced the consumer surplus measure which results in no change in prices or quantities.

```
---- EQU supply

         LOWER          LEVEL          UPPER         MARGINAL

1          .              .            +INF           -1.3821
2          .              .            +INF           -1.3298
3          .              .            +INF           -1.2227
4          .              .            +INF           -1.1282
5          .              .            +INF           -1.2468

---- EQU demand

         LOWER          LEVEL          UPPER         MARGINAL

1          .              .            +INF           -1.5539
2          .              .            +INF           -1.2878
3          .              .            +INF           -1.3783
4          .              .            +INF           -1.3969
5          .              .            +INF           -1.3534
```

Price-Responsive Supply and Demand (QCP Formulation)

```
parameter       eta(i)  Price elasticity of supply from node i;

eta(i) = uniform(0.5, 2);


equation        ssurplus        Social surplus with price elastic supply;


ssurplus..      SURPLUS =e= -sum((i,j), c(i,j) * X(i,j))

        + sum(j,  pref(j) * D(j) * (1 + (1-0.5*D(j)/dref(j)) / epsilon(j)))

        - sum(i,  muref(i) * S(i) * (1 + (0.5*S(i)/sref(i)-1)/eta(i)));


model equilibrium /supply, demand, ssurplus/;

*       Remove the upper bound so as to accommodate price-elasticity:

S.UP(i) = +inf;

solve equilibrium using QCP maximizing SURPLUS;
```

Extending the Marshallian Model

- The Marshallian model fails to account for the *interconnectedness* of supply for goods produced jointly.
- Consider the supply of agricultural land to produce crops. The supply depends on the choice of *crop rotation*
- An increase in supply of one crop (i) may imply increases or decreases in the supply of another crop (j), depending on whether i and j are complements or substitutes in production.
- Aggregate land area varies in the CET model. Extensions of the model are required to incorporate a fixed land supply (some other cost determines decisions over rotations).

Joint Products

- In the American midwest soybeans are alternated with corn because soybeans have nodules on their roots that host bacteria which fix atmospheric nitrogen. Thus, soybeans require less nitrogen to be applied to the field. As a result, an increase in the supply of corn might then increase the supply of soybeans. In short, these are *complementary joint products*.
- In REAP production response to changes in crop prices involves land allocation across alternative *rotations*. The supply of crops is then implicitly determined by the allocation of land to rotations and acreage and crop yields associated with those rotations.
- Consider the allocation of aggregate land ($\mathcal{L}$) to land farmed with one of two rotations r_1 and r_2 For example, technique r_1 may represent a crop rotation of corn and soybeans on alternate years. r_2 might represent a rotation in which corn is planted for two years and soybeans for one.

Joint Products: Application to Land

Land is heterogeneous. Allocation of aggregate land to land under cultivation implies that as land is moved among rotations, the aggregate efficiency changes. Specialization in a more profitable rotation imposes a cost denominated in the ratio of the cultivated land to the aggregate supply.



CET Model: Alternative Values of η

- Low elasticity: $\eta = 0.5$



- High elasticity: $\eta = 2$



Project Independence

Project Independence was an initiative announced by U.S. President Richard Nixon on November 7, 1973, in reaction to the OPEC oil embargo and the resulting 1973 oil crisis. Recalling the Manhattan Project, the stated goal of Project Independence was to achieve energy self-sufficiency for the United States by 1980 through a national commitment to energy conservation and development of alternative sources of energy. Nixon declared that American science, technology and industry could free America from dependence on imported oil (energy independence).

Project Independence

Some of the important initiatives to emerge from Project Independence included lowering highway speeds to 55 mph (89 km/h), converting oil power plants to coal, completion of the Trans-Alaska Pipeline System and diverting federal funds from highway construction to mass transit.

Project Independence

- The PIES initiative was notable for its ambition, the United States has had limited success with energy independence. In 1976 the General Accounting Office reviewed the complex PIES forecasting and analysis system which had been developed by the Federal Energy Administration (FEA) used in preparing the November 1974 Project Independence report.
- The key recommendations of the GAO review were that FEA should add to its plan (i) an analysis of problems resulting from the static nature of the system and the procedures which can be used to alleviate them; (ii) an analysis of the limitations in the environmental impact analysis and the procedures which can be applied to correct them; and (iii) a comprehensive, well-documented verification, validation, and sensitivity testing effort. In spite of these limitations, the analysis represented by the PIES model was well ahead of its time.

The Energy System



Fig. 1. The energy system.

A Framework for Analysis



Fig. 2. The FEA project independence evaluation system.

Curb Your Enthusiasm!

It is well-known that modelers and policy analysts gain access to policymaking arenas based on what they know. Therefore, critics of models are quick to employ various types of technical standards when evaluating policy models in order to assess validity and reliability of claims to knowledge. This article argues that, in the effort to make models better, overreliance on technical standards misses the important political and policy reasons to model: models call attention to the modelers and to their advice about important policy problems of the day. In this sense, models are used as symbols, as claims to authority, whether or not the underlying knowledge is technically up to snuff. Drawing on the experience of energy policy models, this article explores the problem of models as knowledge versus models as symbols and it examines the muddle that conflicts between them produce. (PB Hammond, *Policy Sciences* February 1984, Volume 16, Issue 3, pp 227-243)

A Model of the Model

William Hogan, "Energy Policy Models for Project Independence", *Computers and Operations Research* Vol 2, pp 251–271, 1975.

```
set     j       Consumption regions /j1, j2/

        i       Coal supply regions /i1, i2/

        k       Oil supply regions  /k1, k2/

        r       Refineries          /r1, r2/

        c       Increments for coal /L, M, H/,

        o       Increments for oil  /L, H/,

        p       Energy products         /Light, Heavy, Coal/,

        g(p)    Grades of refined oil / Light, Heavy/;
```

Tables and Figures from the Paper

Table 1. Resource requirements for production levels

	Production (ton/D)	Minimum price/ton($)	New capital/ton	Steel/ton
	0–300	5	1	1
Coal	300–600	6	5	2
region 1	600–1000	8	10	3
	0–200	4	1	1
Coal	200–500	5	5	4
region 2	500–1000	7	6	5



Fig. 3. Steel requirement coal region 1.

Model Variables

```
NONNEGATIVE VARIABLES
        COAL(i,c)       Coal in region i at increment c,
        OIL(k,o)        Oil in region k at increment o,Cent
        XC(i,j)         Coal transport from i to j,
        XO(k,r)         Oil transport from k to refinery r,
        Z(r)            Level of operation in refinery r,
        Y(p,r,j)        Oil products from r to j,
```

Dual Variables

```
P_C(i)          Supply price of coal,
P_O(k)          Supply price of oil,
P_XO(r)         Delivered price of oil,
P_R(g,r)        Price of refinery outputs,
P_D(p,j)        Demand price of products,
P_S             Shadow price of the steel,
P_K             Shadow price of the capital;
```

Equations of the Baby PIES Model

```
oilresource(k)..      sum(o, QO(k,o)) =g= sum(r, XO(k,r));

crudeoil(r)..         sum(k, XO(k,r)) =g= QR(r);

refinedoil(r,g)..     QR(r)*table5(g,r) =g= sum(j, XR(r,g,j));

oildemand(j,g)..      sum(r, XR(r,g,j)) =g= D(g,j);

coalsupply(i)..       sum(c, QC(i,c)) =g= sum(j, XC(i,j));

coaldemand(j)..       sum(i, XC(i,j)) =g= D("coal",j);
```

Equations (cont.)

```
*       Transportation costs:

transportcost..          TCOST =e= sum((r,g,j), XR(r,g,j)*table6(r,j)) +
                                   sum((k,r),   XO(k,r)*table4(k,r)) +
                                   sum((i,j),   XC(i,j)*table2(i,j));
*       Other costs:

othercost..              OCOST =e= sum((k,o), QO(k,o)*table3(k,o,"c0")) +
                                   sum((i,c), QC(i,c)*table1(i,c,"c0")) +
                                   sum(r,     QR(r)*table5("cost",r));

*       First use a simple integrable demand function to define consumer surplus

surplus..        CS =e= sum((p,j), D(p,j)*pref(p) * (1 + (1-D(p,j)/(2*qref(p))/ep

objdef..         OBJ =e= CS - TCOST - OCOST;
```

Equations (cont.)

```
QC.UP(i,c) = table1(i,c,"cap");
QO.UP(k,o) = table3(k,o,"cap");

model pies /all/;
solve pies using QCP maximizing obj;
```